F08JEF (SSTEQR/DSTEQR) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F08JEF (SSTEQR/DSTEQR) computes all the eigenvalues, and optionally all the eigenvectors, of a real symmetric tridiagonal matrix, or of a real symmetric matrix which has been reduced to tridiagonal form.

2 Specification

```
SUBROUTINE F08JEF(COMPZ, N, D, E, Z, LDZ, WORK, INFO)
ENTRY      ssteqr(COMPZ, N, D, E, Z, LDZ, WORK, INFO)
INTEGER           N, LDZ, INFO
real              D(*), E(*), Z(LDZ,*), WORK(*)
CHARACTER*1       COMPZ
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine computes all the eigenvalues, and optionally all the eigenvectors, of a real symmetric tridiagonal matrix T. In other words, it can compute the spectral factorization of T as

$$T = Z\Lambda Z^T,$$

where Λ is a diagonal matrix whose diagonal elements are the eigenvalues λ_i, and Z is the orthogonal matrix whose columns are the eigenvectors z_i. Thus

$$Tz_i = \lambda_i z_i \text{ for } i = 1, 2, \ldots, n.$$

The routine may also be used to compute all the eigenvalues and eigenvectors of a real symmetric matrix A which has been reduced to tridiagonal form T:

$$\begin{aligned} A &= QTQ^T, \text{ where } Q \text{ is orthogonal,} \\ &= (QZ)\Lambda(QZ)^T. \end{aligned}$$

In this case, the matrix Q must be formed explicitly and passed to F08JEF, which must be called with COMPZ = 'V'. The routines which must be called to perform the reduction to tridiagonal form and form Q are:

full matrix	F08FEF (SSYTRD/DSYTRD) + F08FFF (SORGTR/DORGTR)
full matrix, packed storage	F08GEF (SSPTRD/DSPTRD) + F08GFF (SOPGTR/DOPGTR)
band matrix	F08HEF (SSBTRD/DSBTRD) with VECT = 'V'.

F08JEF uses the implicitly shifted QR algorithm, switching between the QR and QL variants in order to handle graded matrices effectively (see Greenbaum and Dongarra [2]). The eigenvectors are normalized so that $\|z_i\|_2 = 1$, but are determined only to within a factor ± 1.

If only the eigenvalues of T are required, it is more efficient to call F08JFF (SSTERF/DSTERF) instead. If T is positive-definite, small eigenvalues can be computed more accurately by F08JGF (SPTEQR/DPTEQR).

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

[2] Greenbaum A and Dongarra J J (1980) Experiments with QR/QL methods for the symmetric triangular eigenproblem *LAPACK Working Note No. 17 (Technical Report CS–89–92)* University of Tennessee, Knoxville

[3] Parlett B N (1980) *The Symmetric Eigenvalue Problem* Prentice–Hall

5 Parameters

1: COMPZ — CHARACTER*1 *Input*

On entry: indicates whether the eigenvectors are to be computed as follows:

if COMPZ = 'N', then only the eigenvalues are computed (and the array Z is not referenced);

if COMPZ = 'I', then the eigenvalues and eigenvectors of T are computed (and the array Z is initialized by the routine);

if COMPZ = 'V', then the eigenvalues and eigenvectors of A are computed (and the array Z must contain the matrix Q on entry).

Constraint: COMPZ = 'N', 'V' or 'I'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix T.

Constraint: N $\geq$ 0.

3: D($*$) — ***real*** array *Input/Output*

Note: the dimension of the array D must be at least max(1,N).

On entry: the diagonal elements of the tridiagonal matrix T.

On exit: the n eigenvalues in ascending order, unless INFO $>$ 0 (in which case see Section 6).

4: E($*$) — ***real*** array *Input/Output*

Note: the dimension of the array E must be at least max(1,N−1).

On entry: the off-diagonal elements of the tridiagonal matrix T.

On exit: the array is overwritten.

5: Z(LDZ,$*$) — ***real*** array *Input/Output*

Note: the second dimension of the array Z must be at least max(1,N) if COMPZ = 'V' or 'I', and at least 1 if COMPZ = 'N'.

On entry: if COMPZ = 'V', Z must contain the orthogonal matrix Q from the reduction to tridiagonal form. If COMPZ = 'I', Z need not be set.

On exit: if COMPZ = 'I' or 'V', the n required orthonormal eigenvectors stored by columns; the ith column corresponds to the ith eigenvalue, where $i = 1, 2, \ldots, n$, unless INFO $>$ 0.

Z is not referenced if COMPZ = 'N'.

6: LDZ — INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08JEF (SSTEQR/DSTEQR) is called.

Constraints:

LDZ $\geq$ 1 if COMPZ = 'N',

LDZ $\geq$ max(1,N) if COMPZ = 'V' or 'I'.

7: WORK(∗) — ***real*** array *Workspace*

Note: the dimension of the array WORK must be at least max(1,2*(N−1)) if COMPZ = 'V' or 'I', and at least 1 if COMPZ = 'N'.

WORK is not referenced if COMPZ = 'N'.

8: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm has failed to find all the eigenvalues after a total of 30×N iterations. In this case, D and E contain on exit the diagonal and off-diagonal elements, respectively, of a tridiagonal matrix orthogonally similar to T. If INFO = i, then i off-diagonal elements have not converged to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $T+E$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the ***machine precision***.

If λ_i is an exact eigenvalue and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq c(n)\epsilon\|T\|_2,$$

where $c(n)$ is a modestly increasing function of n.

If z_i is the corresponding exact eigenvector, and $\tilde{z}_i$ is the corresponding computed eigenvector, then the angle $\theta(\tilde{z}_i, z_i)$ between them is bounded as follows:

$$\theta(\tilde{z}_i, z_i) \leq \frac{c(n)\epsilon\|T\|_2}{\min\limits_{i \neq j} |\lambda_i - \lambda_j|}.$$

Thus the accuracy of a computed eigenvector depends on the gap between its eigenvalue and all the other eigenvalues.

8 Further Comments

The total number of floating-point operations is typically about $24n^2$ if COMPZ = 'N' and about $7n^3$ if COMPZ = 'V' or 'I', but depends on how rapidly the algorithm converges. When COMPZ = 'N', the operations are all performed in scalar mode; the additional operations to compute the eigenvectors when COMPZ = 'V' or 'I' can be vectorized and on some machines may be performed much faster.

The complex analogue of this routine is F08JSF (CSTEQR/ZSTEQR).

9 Example

To compute all the eigenvalues and eigenvectors of the symmetric tridiagonal matrix T, where

$$T = \begin{pmatrix} -6.99 & -0.44 & 0.00 & 0.00 \\ -0.44 & 7.92 & -2.63 & 0.00 \\ 0.00 & -2.63 & 2.34 & -1.18 \\ 0.00 & 0.00 & -1.18 & 0.32 \end{pmatrix}.$$

See also the examples for F08FFF, F08GFF or F08HEF, which illustrate the use of this routine to compute the eigenvalues and eigenvectors of a full or band symmetric matrix.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08JEF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDZ
      PARAMETER        (NMAX=8,LDZ=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, N
*     .. Local Arrays ..
      real             D(NMAX), E(NMAX-1), WORK(2*NMAX-2), Z(LDZ,NMAX)
*     .. External Subroutines ..
      EXTERNAL         ssteqr, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08JEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read T from data file
*
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
*        Calculate all the eigenvalues and eigenvectors of T
*
         CALL ssteqr('I',N,D,E,Z,LDZ,WORK,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.GT.0) THEN
            WRITE (NOUT,*) 'Failure to converge.'
         ELSE
*
*           Print eigenvalues and eigenvectors
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (D(I),I=1,N)
            WRITE (NOUT,*)
            IFAIL = 0
*
            CALL X04CAF('General',' ',N,N,Z,LDZ,'Eigenvectors',IFAIL)
*
         END IF
      END IF
      STOP
*
99999 FORMAT (3X,(8F8.4))
      END
```

9.2 Program Data

```
F08JEF Example Program Data
  4                            :Value of N
 -6.99   7.92   2.34   0.32
 -0.44  -2.63  -1.18           :End of matrix T
```

9.3 Program Results

```
 F08JEF Example Program Results

 Eigenvalues
    -7.0037 -0.4059  2.0028  8.9968

 Eigenvectors
          1       2       3       4
 1   0.9995 -0.0109 -0.0167 -0.0255
 2   0.0310  0.1627  0.3408  0.9254
 3   0.0089  0.5170  0.7696 -0.3746
 4   0.0014  0.8403 -0.5397  0.0509
```